Lab 01: SPICE

Part (1)

(1)

```
Voltage Divider Netlist
* Any text after the asterisk '*' is ignored by SPICE
* Voltage Divider
V1 1 0 12
R1 1 2 1k
R2 2 0 2k
* Perform operating point analysis
*** add line here ***
.OP
*** add line here ***
.END
```

```
Voltage Divider Netlist
       --- Operating Point ---

V(1):          12          voltage
V(2):          8           voltage
I(R2):         0.004       device_current
I(R1):         0.004       device_current
I(V1):         -0.004      device_current
```

(2)

```
Simple RC Circuit

* Circuit Description

* Parameters
*** add line here ***
.PARAM CPAR 500p
* Signal sources
*** complete this line *** (0V 1V 0 100n 100n 10u 20u)
Vtran 1 0 PULSE(0V 1V 0 100n 100n 10u 20u)
* Circuit elements
R1 1 2 1k
C1 2 0 {CPAR}

* Initial conditions
* Analysis request
* Run transient for 40us with 100ns step
*** add line here ***
.TRAN 100ns 40us
* Use parametric sweep for CPAR: 500p:500p:1.5n
*** add line here ***
.STEP PARAM CPAR 500p 1.5n 500p
* Measure rise time from 10% to 90%
.MEAS TRAN TRISE
+ TRIG when v(2) = 0.1 CROSS = 1
*** add line here ***
+ TARG when v(2) = 0.9 CROSS = 1
*** add line here ***
.END
```




```
Circuit: Simple RC Circuit

.OP point found by inspection.
.step cpar=5e-010
.step cpar=1e-009
.step cpar=1.5e-009


Measurement: trise
  step      trise FROM  TO
     1      1.09912e-006      1.03619e-007      1.20274e-006
     2      2.19839e-006      1.56143e-007      2.35454e-006
     3      3.29638e-006      2.08794e-007      3.50517e-006
```

(3)

```
Simple RC Circuit

* Circuit Description

* Parameters
*** add line here ***
.PARAM CPAR 500p
* Signal sources
*** add line here ***
Vac 1 0 AC 1V
* Circuit elements
R1 1 2 1k
C1 2 0 {CPAR}

* Analysis request
* Run ac sweep from 1Hz to 100MEG with 10 pts per decade
*** add line here ***
.AC DEC 10 1Hz 100MEG
* Use parametric sweep for CPAR: 500p:500p:1.5n
*** add line here ***
.STEP PARAM CPAR 500p 1.5n 500p
* Output request
.PRINT AC V(1) V(2)
.PLOT AC V(1) V(2)

* Measure the peak
.MEAS AC PEAK max mag(V(2))

* Measure bandwidth using PEAK/sqrt(2)
.MEAS AC BW
*** add line here ***
+ WHEN mag(V(2)) = PEAK/sqrt(2)
*** add line here ***
.END
```



```
Circuit: Simple RC Circuit

.step cpar=5e-010
.step cpar=1e-009
.step cpar=1.5e-009


Measurement: peak
  step      MAX(mag(v(2)))      FROM   TO
     1      (-4.28644e-011dB,0°)    1      1e+008
     2      (-1.71454e-010dB,0°)    1      1e+008
     3      (-3.8577e-010dB,0°)     1      1e+008

Measurement: bw
  step      mag(v(2))=peak/sqrt(2)
     1      318461
     2      159204
     3      106411
```

Part (2)

(1)

```
* Non-ideal Op-amp (small signal) Subcircuit
*
* Subcircuit Description
*
.SUBCKT small_signal_OPAMP IN+ IN- OUT
* connections: |   |   |
*          +ve input   |   |
*                  -ve input   |
*                                  output
Ginput 0 4 IN+ IN- 10
Iopen1 IN+ 0 0A                    ; redundant connection made at +ve input terminal
Iopen2 IN- 0 0A                    ; redundant connection made at -ve input terminal
R1 4 0 1k
C1 4 0 159.155n
Eoutput OUT 0 4 0 1
.ENDS small_signal_OPAMP
.END
```

<u>Circuit parameters calculations:</u>

$A_o\omega_B = \omega_t \rightarrow (10^4)f_B = 10MHz \rightarrow f_B = 1$ kHz

$A_o = G_mR_1 = 10^4 \rightarrow (1)$

$\omega_B = 2\pi f_B = 1/R_1C_1 \rightarrow 2\pi(1\text{ kHz}) = 1/R_1C_1 \rightarrow (2)$

Let $R_1 = 1$ KΩ, then solve (1) & (2) together:

C_1 = 159.155 nF, G_m= 10

(2)

```
Non-inverting Amplifier
*
* Subcircuit Description
*
.INC small_signal_OPAMP.cir

*
* Circuit Description
*

* Signal source
Vin 3 0 DC 1

* Circuit elements
R1 0 2 1k
Rf 2 1 9k
XOPAMP1 3 2 1 small_signal_OPAMP

*
* Analysis Request
*
.TF V(1) Vin

.END
```

```
Non-inverting Amplifier
       --- Transfer Function ---

Transfer_function:          9.99001       transfer
vin#Input_impedance:        1e+020        impedance
output_impedance_at_V(1):   0             impedance
```

An ideal op-amp has:

- an infinite input impedance (shown in the output 1e+020 referring to a very high input impedance)
- Output impedance equals zero (as shown in the output)

The gain of the non-inverting amplifier is $A_v = 1+R_f/R_1 = 1+9/1 = 10$ for ideal op-amp.

For non-ideal op-amp, $A_v = (1+R_f/R_1)(1-(1+R_f/R_1)/A_0) = 9.99$ (as shown in the output).

(3)



(V_sig in green and V_out in blue)

```
Non-inverting Amplifier
*
* Subcircuit Description
*
.INC small_signal_OPAMP.cir

*
* Circuit Description
*
* Parameters
.PARAM PERIOD = 1m

* Signal source
Vin 3 0 SIN(0 1 1k 0 0)

* Circuit elements
R1 0 2 1k
Rf 2 1 9k
XOPAMP1 3 2 1 small_signal_OPAMP

*
* Analysis Request
*
.TRAN {PERIOD/50} {2*PERIOD}

* Output request
.PRINT TRAN V(1) V(2) V(3)
.PLOT TRAN V(1) V(2) V(3)

* Measure the peak
.MEAS TRAN Vout_PEAK max ABS(V(1))
.MEAS TRAN Vsig_PEAK_+node max ABS(V(3))
.MEAS TRAN Vsig_PEAK_-node max ABS(V(2))

.END
```

```
vout_peak: MAX(abs(v(1)))=9.98038 FROM 0 TO 0.002
vsig_peak_+node: MAX(abs(v(3)))=0.999081 FROM 0 TO 0.002
```

(4)

Voltage gain = 9.98038/0.999081 = 9.98956

Hand analysis	TF analysis	TRAN analysis
9.99	9.99001	9.98956

The TF is almost the same as the hand analysis. In contrast, the TRAN analysis is lower because it uses a frequency 1kHz, and the gain decreases with frequency.

(5)




```
differential_input_peak: MAX(abs(v(3)-v(2)))=0.00143945 FROM 0 TO 0.002
```

differential input waveform

The amplitude of the differential input is 0.999081-0.998038 = 1mV because $V_{in}= (V_{out})/(A_o/\sqrt{2})= 10/(10^4/\sqrt{2})= 1.414mV$

(6)




```
Non-inverting Amplifier
*
* Subcircuit Description
*
.INC small_signal_OPAMP.cir

*
* Circuit Description
*
* Parameters
*.PARAM PERIOD = 1m
.PARAM PERIOD = 100n ; For input frequency equal to the UGF

* Signal source
*Vin 3 0 SIN(0 1 1k 0 0)
Vin 3 0 SIN(0 1 10MEG 0 0) ; Input frequency equal to the UGF
* Circuit elements
R1 0 2 1k
Rf 2 1 9k
XOPAMP1 3 2 1 small_signal_OPAMP

*
* Analysis Request
*
.TRAN {PERIOD/50} {8*PERIOD}

* Output request
.PRINT TRAN V(1) V(2) V(3)
.PLOT TRAN V(1) V(2) V(3)

* Measure the peak
.MEAS TRAN Vout_PEAK max ABS(V(1))
.MEAS TRAN Vsig_PEAK_+node max ABS(V(3))
.MEAS TRAN Vsig_PEAK_-node max ABS(V(2))
.MEAS TRAN Differential_input_PEAK max ABS(V(3)-V(2))
.END
```

SPICE Error Log: D:\Summer2022\AMS Course\ams_labs_pdf\lab1\non_inverting_amplifier_with_sine_s

```
Circuit: Non-inverting Amplifier

.OP point found by inspection.

vout_peak: MAX(abs(v(1)))=1.72696 FROM 0 TO 8e-007
vsig_peak_+node: MAX(abs(v(3)))=0.999884 FROM 0 TO 8e-007
vsig_peak_-node: MAX(abs(v(2)))=0.172696 FROM 0 TO 8e-007
differential_input_peak: MAX(abs(v(3)-v(2)))=1.05709 FROM 0 TO 8e-007


Date: Fri Aug 05 23:33:24 2022
Total elapsed time: 0.275 seconds.

tnom = 27
temp = 27
method = modified trap
totiter = 2082
traniter = 2082
tranpoints = 1042
accept = 1042
rejected = 0
matrix size = 6
fillins = 1
solver = Normal
Matrix Compiler1: 182 bytes object code size  0.2/0.2/[0.1]
Matrix Compiler2: off  [0.1]/0.4/0.2
```

Differential input waveform:



The amplitude of the **differential** input signal is 1V because at UGF the gain is 1, so because $V_{in} = V_{out}/A_o = 1/1 = 1V$

Note that the reported peak value of the output signal is incorrect due to the dc shift that fades with time.

(7)




```
Non-inverting Amplifier
*
* Subcircuit Description
*
.INC small_signal_OPAMP.cir

*
* Circuit Description
*
* Parameters
.PARAM RF_PAR 9k

* Signal source
Vac 3 0 AC 1V

* Circuit elements
R1 0 2 1k
Rf 2 1 {RF_PAR}
XOPAMP1 3 2 1 small_signal_OPAMP

*
* Analysis Request
*
.AC DEC 10 1Hz 100MEG
.STEP PARAM RF_PAR LIST 9k 4k

* Output request
.PRINT TRAN V(1) V(2) V(3)
.PLOT TRAN V(1) V(2) V(3)

* Measure the peak
.MEAS AC DC_GAIN max mag(V(1))
.MEAS AC CUTOFF_FREQ WHEN mag(V(1)) = DC_GAIN/sqrt(2)
.MEAS AC UGF WHEN mag(V(1)) = 1

.END
```

```
.step rf_par=4000
.step rf_par=9000


Measurement: dc_gain
  step       MAX(mag(v(1)))     FROM   TO
     1        (13.9751dB,0°)     1      1e+008
     2        (19.9913dB,0°)     1      1e+008

Measurement: cutoff_freq
  step       mag(v(1))=dc_gain/sqrt(2)
     1        2.00142e+006
     2        1.00107e+006

Measurement: ugf
  step       mag(v(1))=1
     1        9.83357e+006
     2        9.95973e+006
```

(8)

If you increase the input amplitude in AC analysis and transient analysis, do you expect to see clipping in the output? Why?

No, because our model doesn't include this large-signal non-ideality.

(9)

For $R_f = 9k$

	DC gain	Cutoff frequency	UGF
Hand analysis	10 (assuming ideal op-amp) 9.99 (taking the gain error into account)	UGF/DC_Gain = 1MHz	= Closed-loop UGF = open-loop UGF = 10MHz

AC analysis	19.9913dB =9.989989	1MHz	9.95973MHz

For $R_f = 4k$

	DC gain	Cutoff frequency	UGF
Hand analysis	5(assuming ideal op-amp)	UGF/DC_Gain = 2MHz	= Closed-loop UGF = open-loop UGF = 10MHz
AC analysis	13.9751dB = 4.9975	2MHz	9.83357MHz

Comment: the closed-loop amplifier decrease the dc gain and increases the cutoff frequency such that the gain-bandwidth product is the same, so by changing the feedback resistance, the DC gain and cutoff frequency change while UGF is the same.



Figure 8.37 Frequency response of open-loop op amp and closed-loop circuit.